Brian Walsh
September 17-18, 2005

Menus and displays:

GO



MORE



SOLV



BRian's dUCT analysis
for HP 48G/GX/49G/49g+

The purpose of this program is obvious: to analyze and size ductwork for HVAC systems.
The objective of this presentation is to demonstrate an easy to use (for me anyway) system for input and output of variable values, with a supporting menu system and without requiring tedious unnecessary keystrokes.

The application (in the SOLV routine, listed below) uses the built in DARCY function which deserves more attention than it seems to have received, as it solves for the Darcy friction factor (an iterative solution since the variable cannot be isolated) much more quickly than if using the conventional ROOT function or other methods. Those familiar with fluid mechanics, used in mechanical, civil, and chemical engineering and probably several other fields, are familiar with the Moody diagram used for manually determining the friction factor. A copy of such a diagram is included after the program listings for reference. Since the flow is always in the turbulent region (where Reynolds number Re > 10,000) and the lines in the diagram slope from the upper left to the lower right, one can solve for the friction factor f by successive approximations (using the previous result or initial guess), yet it's interesting that using the DARCY function proved to be quite a bit faster than even that method. Hence, use of the DARCY function (or its sibling, FANNING) is strongly recommended for efficient, rapid solutions.

Also included at the end of this document is a screen capture of a spreadsheet I created which returns identical results as the BRUCT routines. Formulas and references are given. Anyone wanting a copy of this should please email me.

To use:
In BRUCT directory, press GO which will display the current values of the inputs.
Press INIT to set default values of length, density, dynamic viscosity, and absolute roughness.
Input CFM and press (unshifted) CFM menu key. Updated input values are displayed.
Do similarly for W and H. If the duct is round, input diameter for W and 0 (zero) for H.
Press MORE to change length, density, dynamic viscosity, and absolute roughness if desired. To return to entering CFM, W, and H press the BACK menu key.
Press SOLV to compute and display equivalent diameter, velocity, velocity pressure, friction, Darcy friction factor, Reynolds number, and hydraulic diameter. You can recall the values of De, VEL, VP, and FRIC by pressing their (unshifted) menu keys. To change inputs, press BACK.
You can recall a current input value by pressing the corresponding left or right shifted menu key.
Input a different value or just press the menu key to continue with the current value.

Future enhancements will include selectable units, probably using the CHOOSE function. This could give additional flexibility to these routines and enable solutions for other fluids such as water, compressed air, steam, and so on, not to mention incorporating metric conversions on the fly.

Inputs:

Name	Description	Units	Initialized value
CFM	airflow	ft^3/min	none
W	duct width (or dia.)	in	none
H	duct height (or 0 if round)	in	none
L	duct length	ft	100.
ρ	density	lb/ ft^3	0.075
MU	dynamic viscosity	centipoise	0.0181
∈	absolute roughness	ft	0.0005

Outputs:

De	equivalent diameter	in
VEL	velocity	ft/min
VP	velocity pressure	inH2O
FRIC	friction (pressure loss)	inH2O
f	Darcy friction factor	none
Re	Reynolds number	none
Dh	hydraulic diameter	in

BRUCT

Bytes: 1760.5 Chk: #19CBh

```
%%HP: T(3)A(D)F(.);
DIR
```

GO

```
« { { "INIT" {
  « INIT GO
  » } } { "CFM" {
  « 'CFM' STO GO
  » CFM CFM } } { "W" {
  « 'W' STO GO
  » W W } } { "H" {
  « 'H' STO GO
  » H H } } MORE SOLV } SHOIN TMENU
»
```

Bytes: 255 Chk: #E05Ah

CFM

```
150000.
```

SHO

```
« "{var1 var2 ...} →" DROP 1.
  « DUP
    IFERR RCL
    THEN DROP "Undef"
    END SWAP →TAG
  » DOLIST "
" ADD ΣLIST CLLCD 1. DISP 3. FREEZE
»
```

Bytes: 127 Chk: #5D85h

SHOIN

```
« { L CFM W H ρ MU ε } SHO
»
```

Bytes: 65.5 Chk: #E726h

MORE

```
« { { "L" {
  « 'L' STO MORE
  » L L } } { "ρ" {
  « 'ρ' STO MORE
  » ρ ρ } } { "MU" {
  « 'MU' STO MORE
  » MU MU } } { "ε" {
  « 'ε' STO MORE
  » ε ε } } { "BACK" { GO } } SOLV } SHOIN TMENU
»
```

Bytes: 288.5 Chk: #A191h

SOLV

```
« H '1.3*(W*H)^.625/(W+H)^.25' W IFTE 'De' STO
  H 'W*H/144.' 'π*W^2./576.' IFTE DUP 'A' STO
  CFM SWAP / DUP 'VEL' STO 1097. / SQ ρ * 'VP' STO
  H '4.*A/(2.*(W+H)/144.)' W IFTE 'Dh' STO
  2.06689 ρ * MU / Dh * VEL * DUP 'Re' STO
  ε 12. * Dh / SWAP DARCY DUP 'f' STO
  12. * L * Dh / VP * 'FRIC' STO OUTPUT
»
```

Bytes: 497 Chk: #A007h

INIT

```
« .075 'ρ' STO .0181 'MU' STO .0005 'ε' STO 100. 'L' STO
»
```

Bytes: 109.5 Chk: #FF68h

OUTPUT

```
« { GO } ORDER { De VEL VP FRIC f Re Dh } SHO { De VEL VP FRIC { "BACK"
{ GO } } SOLV } TMENU
»
```

Bytes: 150 Chk: #7B34h

FRIC .14745263973

f 1.26970327733E-2

Re 1831607.5707

Dh 66.5346534654

VP .643897834384

VEL 3214.28571428

A 46.6666666667

De 85.0572437807

H 42.

W 160.

L 100.

ϵ .0005

MU .0181

ρ .075

END

2.8 2005 ASHRAE Handbook—Fundamentals



Fig. 13 Relation Between Friction Factor and Reynolds Number
(Moody 1944)



Brian Walsh
bpwalsh@speakeasy.net
847.382.8433

.END. REG 831